# Archlive <http://archlive-pkg.googlecode.com>

pkgbase=linux-cjk
pkgname=('linux-cjk' 'linux-headers' ) # 'linux-docs') # Build stock -ARCH kernel
pkgname=linux-cjk       # Build kernel with a different name
_kernelname=${pkgname#linux}
_basekernel=3.0
pkgver=${_basekernel}.1
pkgrel=1
arch=('i686' 'x86_64')
url="http://www.kernel.org/"
license=('GPL2')
#makedepends=('xmlto' 'docbook-xsl')
options=('!strip')
source=("ftp://ftp.kernel.org/pub/linux/kernel/v3.0/linux-${_basekernel}.tar.bz2"
         "ftp://ftp.kernel.org/pub/linux/kernel/v3.0/patch-${pkgver}.bz2"
        # the main kernel config files
        'config' 'config.x86_64'
        # standard config files for mkinitcpio ramdisk
        "${pkgname}.preset"
        'fix-i915.patch'
        'change-default-console-loglevel.patch'
	# Gentoo patches
	#http://sources.gentoo.org/cgi-bin/viewvc.cgi/linux-patches/genpatches-2.6/trunk/3.0/4200_fbcondecor-0.9.6.patch
	4200_fbcondecor-0.9.6-1.patch
	# aufs2.1 friendly patches    (from http://git.c3sl.ufpr.br/pub/scm/aufs/aufs2-standalone.git)
	aufs2-base.patch
	#aufs2-kbuild.patch
	aufs2-standalone.patch
	#loopback.patch
	#proc_map.patch
	#aufs2.1-3-files.patch # by manual
	### Chinese Characters Patches 孙海勇 大侠的 内核中文补丁  http://youbest.cublog.cn/
	#http://zdbr.net.cn/download/utf8-kernel-2.6.29-core-1.patch.bz2
	utf8-kernel-3.0-core-1.patch
	http://zdbr.net.cn/download/utf8-kernel-2.6-fonts-3.patch.bz2
	http://zdbr.net.cn/download/utf8-kernel-2.6.31-fbcondecor-1.patch.bz2
)
md5sums=('398e95866794def22b12dfbc15ce89c0'
         'ac49f7907f1fc85fbab92d0f1aa1552a'
         'eb6813c26d41728a1bc681eee9957f9c'
         '5795aa406963b8d95156613a2110c9aa'
         'eb14dcfd80c00852ef81ded6e826826a'
         '263725f20c0b9eb9c353040792d644e5'
         '9d3c56a4b999c8bfbd4018089a62f662'
         '865216ca79342c67dc3aa43c18245694'
         '2f4710f4571de355445cb79bdf9f7a30'
         'da6650176eb2c5b0e3d5c88d5a645c59'
         'f1fcfa8f4f27e57552b58ced84ac3b9f'
         'f3271bc1801fadfa2f324f5547b16ceb'
         'cebee1045a00b7632223807f799f9823')

_build_dir=${srcdir}/linux-3.0.1

prepare() {
	cd ${_build_dir}
	#patch -p1 -i "${srcdir}/patch-${pkgver}"

	# add latest fixes from stable queue, if needed
	# http://git.kernel.org/?p=linux/kernel/git/stable/stable-queue.git

	msg2 "fix 19234 i1915 display size"
	patch -Np1 -i "${startdir}/fix-i915.patch"

	msg2 "set DEFAULT_CONSOLE_LOGLEVEL to 4 (same value as the 'quiet' kernel param)"
	# remove this when a Kconfig knob is made available by upstream
	# (relevant patch sent upstream: https://lkml.org/lkml/2011/7/26/227)
	patch -Np1 -i "${startdir}/change-default-console-loglevel.patch"

	msg2 "add Gentoo fbcondecor patch"
	patch -Np1 -i "${startdir}/4200_fbcondecor-0.9.6.patch"

	msg2 "Add the kernel Chinese characters patches 加入孙海勇 大侠的 内核中文补丁  http://youbest.cublog.cn/"
	patch -Np1 -i $srcdir/utf8-kernel-3.0-core-1.patch
	patch -Np1 -i $srcdir/utf8-kernel-2.6.31-fbcondecor-1.patch
	patch -Np1 -i $srcdir/utf8-kernel-2.6-fonts-2.patch

	msg2 "Add aufs2.1 patch sets"
	#patch -Np1 -i $startdir/aufs2.1-3-files.patch
	patch -Np1 -i $startdir/aufs2-base.patch
	#patch -Np1 -i $startdir/aufs2-kbuild.patch
	patch -Np1 -i $startdir/aufs2-standalone.patch
	#patch -Np1 -i $startdir/loopback.patch
	#patch -Np1 -i $startdir/proc_map.patch

	if [ "${CARCH}" = "x86_64" ]; then
		cat ${startdir}/config.x86_64 > ./.config
	else
		cat ${startdir}/config > ./.config
	fi

	if [ "${_kernelname}" != "" ]; then
		sed -i "s|CONFIG_LOCALVERSION=.*|CONFIG_LOCALVERSION=\"${_kernelname}\"|g" ./.config
	fi

	# remove the sublevel from Makefile
	# this ensures our kernel version is always 3.X-ARCH
	# this way, minor kernel updates will not break external modules
	# we need to change this soon, see FS#16702
	sed -ri 's|^(SUBLEVEL =).*|\1|' Makefile

	# get kernel version
	make prepare

	# load configuration
	# Configure the kernel. Replace the line below with one of your choice.
	#make menuconfig # CLI menu for configuration
	#make nconfig # new CLI menu for configuration
	#make xconfig # X-based configuration
	#make oldconfig # using old config from previous kernel version
	# ... or manually edit .config

	####################
	# stop here
	# this is useful to configure the kernel
	#msg "Stopping build"
	#return 1
	####################
	yes "" | make config
}

build() {

	prepare
	cd "${_build_dir}"
	# build!
	make ${MAKEFLAGS} bzImage modules
}

package_linux-cjk() {
	pkgdesc="The Linux Kernel and modules, with CJK patch set"
	groups=('base')
	depends=('coreutils' 'module-init-tools>=3.16' 'mkinitcpio>=0.7')
	optdepends=('crda: to set the correct wireless channels of your country')
	provides=('kernel26' "linux")
	#conflicts=('kernel26')
	#replaces=('kernel26')
	backup=("etc/mkinitcpio.d/${pkgname}.preset")
	install=${pkgname}.install

	cd "${_build_dir}"

	KARCH=x86

	# get kernel version
	_kernver="$(make kernelrelease)"

	mkdir -p "${pkgdir}"/{lib/modules,lib/firmware,boot}
	make INSTALL_MOD_PATH="${pkgdir}" modules_install
	cp arch/$KARCH/boot/bzImage "${pkgdir}/boot/vmlinuz-${pkgname}"

	# add vmlinux
	install -D -m644 vmlinux "${pkgdir}/usr/src/linux-${_kernver}/vmlinux"

	# install fallback mkinitcpio.conf file and preset file for kernel
	install -D -m644 "${srcdir}/${pkgname}.preset" "${pkgdir}/etc/mkinitcpio.d/${pkgname}.preset"

	# set correct depmod command for install
	sed \
		-e  "s/KERNEL_NAME=.*/KERNEL_NAME=${_kernelname}/g" \
		-e  "s/KERNEL_VERSION=.*/KERNEL_VERSION=${_kernver}/g" \
		-i "${startdir}/${pkgname}.install"
	sed \
		-e "s|default_image=.*|default_image=\"/boot/initramfs-${pkgname}.img\"|g" \
		-e "s|fallback_image=.*|fallback_image=\"/boot/initramfs-${pkgname}-fallback.img\"|g" \
		-e "s|ALL_kver=.*|ALL_kver=\"/boot/vmlinuz-${pkgname}\"|g" \
		-i "${pkgdir}/etc/mkinitcpio.d/${pkgname}.preset"

	# remove build and source links
	rm -f "${pkgdir}"/lib/modules/${_kernver}/{source,build}
	# remove the firmware
	rm -rf "${pkgdir}/lib/firmware"
	# gzip -9 all modules to safe 100MB of space
	find "${pkgdir}" -name '*.ko' -exec gzip -9 {} \;
}

package_linux-headers() {
	pkgdesc="Header files and scripts for building modules for linux kernel"
	provides=('kernel26-headers')
	conflicts=('kernel26-headers')
	replaces=('kernel26-headers')

	mkdir -p "${pkgdir}/lib/modules/${_kernver}"

	cd "${pkgdir}/lib/modules/${_kernver}"
	ln -sf ../../../usr/src/linux-${_kernver} build

	cd "${_build_dir}"
	install -D -m644 Makefile \
		"${pkgdir}/usr/src/linux-${_kernver}/Makefile"
	install -D -m644 kernel/Makefile \
		"${pkgdir}/usr/src/linux-${_kernver}/kernel/Makefile"
	install -D -m644 .config \
		"${pkgdir}/usr/src/linux-${_kernver}/.config"

	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/include"

	for i in acpi asm-generic config crypto drm generated linux math-emu \
		media net pcmcia scsi sound trace video xen; do
		cp -a include/${i} "${pkgdir}/usr/src/linux-${_kernver}/include/"
	done

	# copy arch includes for external modules
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/arch/x86"
	cp -a arch/x86/include "${pkgdir}/usr/src/linux-${_kernver}/arch/x86/"

	# copy files necessary for later builds, like nvidia and vmware
	cp Module.symvers "${pkgdir}/usr/src/linux-${_kernver}"
	cp -a scripts "${pkgdir}/usr/src/linux-${_kernver}"

	# fix permissions on scripts dir
	chmod og-w -R "${pkgdir}/usr/src/linux-${_kernver}/scripts"
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/.tmp_versions"

	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/arch/${KARCH}/kernel"

	cp arch/${KARCH}/Makefile "${pkgdir}/usr/src/linux-${_kernver}/arch/${KARCH}/"

	if [ "${CARCH}" = "i686" ]; then
		cp arch/${KARCH}/Makefile_32.cpu "${pkgdir}/usr/src/linux-${_kernver}/arch/${KARCH}/"
	fi

	cp arch/${KARCH}/kernel/asm-offsets.s "${pkgdir}/usr/src/linux-${_kernver}/arch/${KARCH}/kernel/"

	# add headers for lirc package
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/video"

	cp drivers/media/video/*.h  "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/video/"

	for i in bt8xx cpia2 cx25840 cx88 em28xx et61x251 pwc saa7134 sn9c102; do
		mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/video/${i}"
		cp -a drivers/media/video/${i}/*.h "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/video/${i}"
	done

	# add docbook makefile
	install -D -m644 Documentation/DocBook/Makefile \
		"${pkgdir}/usr/src/linux-${_kernver}/Documentation/DocBook/Makefile"

	# add dm headers
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/drivers/md"
	cp drivers/md/*.h "${pkgdir}/usr/src/linux-${_kernver}/drivers/md"

	# add inotify.h
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/include/linux"
	cp include/linux/inotify.h "${pkgdir}/usr/src/linux-${_kernver}/include/linux/"

	# add wireless headers
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/net/mac80211/"
	cp net/mac80211/*.h "${pkgdir}/usr/src/linux-${_kernver}/net/mac80211/"

	# add dvb headers for external modules
	# in reference to:
	# http://bugs.archlinux.org/task/9912
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/dvb/dvb-core"
	cp drivers/media/dvb/dvb-core/*.h "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/dvb/dvb-core/"
	# and...
	# http://bugs.archlinux.org/task/11194
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/include/config/dvb/"
	cp include/config/dvb/*.h "${pkgdir}/usr/src/linux-${_kernver}/include/config/dvb/"

	# add dvb headers for http://mcentral.de/hg/~mrec/em28xx-new
	# in reference to:
	# http://bugs.archlinux.org/task/13146
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/dvb/frontends/"
	cp drivers/media/dvb/frontends/lgdt330x.h "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/dvb/frontends/"
	cp drivers/media/video/msp3400-driver.h "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/dvb/frontends/"

	# add dvb headers
	# in reference to:
	# http://bugs.archlinux.org/task/20402
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/dvb/dvb-usb"
	cp drivers/media/dvb/dvb-usb/*.h "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/dvb/dvb-usb/"
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/dvb/frontends"
	cp drivers/media/dvb/frontends/*.h "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/dvb/frontends/"
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/common/tuners"
	cp drivers/media/common/tuners/*.h "${pkgdir}/usr/src/linux-${_kernver}/drivers/media/common/tuners/"

	# add xfs and shmem for aufs building
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/fs/xfs"
	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}/mm"
	cp fs/xfs/xfs_sb.h "${pkgdir}/usr/src/linux-${_kernver}/fs/xfs/xfs_sb.h"

	# copy in Kconfig files
	for i in `find . -name "Kconfig*"`; do
		mkdir -p "${pkgdir}"/usr/src/linux-${_kernver}/`echo ${i} | sed 's|/Kconfig.*||'`
		cp ${i} "${pkgdir}/usr/src/linux-${_kernver}/${i}"
	done

	chown -R root.root "${pkgdir}/usr/src/linux-${_kernver}"
	find "${pkgdir}/usr/src/linux-${_kernver}" -type d -exec chmod 755 {} \;

	# strip scripts directory
	find "${pkgdir}/usr/src/linux-${_kernver}/scripts" -type f -perm -u+w 2>/dev/null | while read binary ; do
		case "$(file -bi "${binary}")" in
			*application/x-sharedlib*) # Libraries (.so)
				/usr/bin/strip ${STRIP_SHARED} "${binary}";;
			*application/x-archive*) # Libraries (.a)
				/usr/bin/strip ${STRIP_STATIC} "${binary}";;
			*application/x-executable*) # Binaries
				/usr/bin/strip ${STRIP_BINARIES} "${binary}";;
		esac
	done

	# remove unneeded architectures
	rm -rf "${pkgdir}"/usr/src/linux-${_kernver}/arch/{alpha,arm,arm26,avr32,blackfin,cris,frv,h8300,ia64,m32r,m68k,m68knommu,mips,microblaze,mn10300,parisc,powerpc,ppc,s390,sh,sh64,sparc,sparc64,um,v850,xtensa}
}

package_linux-docs() {
	pkgdesc="Kernel hackers manual - HTML documentation that comes with the Linux kernel."
	provides=('kernel26-docs')
	conflicts=('kernel26-docs')
	replaces=('kernel26-docs')

	cd "${_build_dir}"

	mkdir -p "${pkgdir}/usr/src/linux-${_kernver}"
	mv Documentation "${pkgdir}/usr/src/linux-${_kernver}"
	find "${pkgdir}" -type f -exec chmod 444 {} \;
	find "${pkgdir}" -type d -exec chmod 755 {} \;

	# remove a file already in linux package
	rm -f "${pkgdir}/usr/src/linux-${_kernver}/Documentation/DocBook/Makefile"
}
